



HP-35s CALCULATOR PROGRAMS

Mn/DOT Office of Land Management

Surveys Research & Support Unit



General Instructions for Keyboard Entry

The listings that follow show the **KEYSTROKES** (the text on the face of the keys to press) and **DISPLAY** (what should appear on the display after each step is entered) for entering the programs. You can step forward and backward through the program to check or edit entries with the cursor keys.

⬅ is the yellow shift key.

➡ is the blue shift key.

↵ is the white arrow key to the far right of the ENTER key.

ⓐ is the “ON” key in the lower left corner of the keyboard.

⬅ ⬆ ⬇ ⬅ are the four silver cursor keys in the upper right corner of the keyboard. These keys scroll up and down through program listings or left and right on program lines.

The user does not have to use the ⓐ key to exit program mode after entering a program if another program is to be entered immediately. Simply enter the next program label and continue. The order in which labels are entered is irrelevant. Most of the programs require the **UTILITY SUBROUTINES** (LBL U) and/or the **NUMERIC AND PROMPTING DATA** (LBL W) to be entered into the HP-35s.

Verifying Correct Program Entry

After entering a program and exiting program mode, the user can verify the correct entry of the program by accessing its checksum and comparing it to the checksum in the listings.

To find the checksum for any label, key in **MEM** {⬅ ⬇ 2} to display the program catalog. Scroll to the label to be checked using the cursor keys. Key in **SHOW** {⬅ ENTER} to display the label’s checksum. Hold down the ENTER key to keep the checksum visible.

Please note that the checksum for LBL W may not match this document. See page 47.

Contact

Roy Graff
395 John Ireland Blvd, MS 643
Saint Paul, MN 55155-1899

Phone: (651) 366-3443
Fax: (651) 366-3450
E-mail: roy.graff@dot.state.mn.us

Non-Mn/DOT users can expect only limited support. Please report program or listing errors.

© Minnesota Department of Transportation, 2006-2010

Warning: The user releases the Minnesota Department of Transportation from all liability resulting from inaccuracies in these application listings.

INVERSE TRAVERSE PROGRAM RADIAL INVERSE PROGRAM

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
→ XEQ 7	R001 LBL R	
← ▴ 1 2	R002 SF 2	{Flag 2 set = radial inverse mode}
GTO COS 0 0 3	R003 GTO I003	
→ XEQ COS	I001 LBL I	
← ▴ 2 2	I002 CF 2	{Flag 2 clear = inverse traverse mode}
← ▴ 1 4	I003 SF 4	
← ▴ 1 0	I004 SF 0	
XEQ 4 ENTER	I005 XEQ U001	{Get beginning point}
← ▴ 2 0	I006 CF 0	
XEQ 4 ENTER	I007 XEQ U001	{Get next point}
RCL R↓	I008 RCL E	
RCL - 1	I009 RCL- X	
RCL +/-	I010 RCL N	
RCL - 2	I011 RCL- Y	
XEQ 4 0 8 5	I012 XEQ U085	{To Polar}
→ RCL R/S	I013 STO A	
x◀▶y	I014 x◁y	
→ RCL MODE	I015 STO D	
← R↓ MODE	I016 VIEW D	{Read inverse distance}
← R↓ R/S	I017 VIEW A	{Read azimuth in D.MMSSss}
← ▴ 3 2	I018 FS? 2	{Flag 2 set = radial mode}
GTO COS 0 0 7	I019 GTO I007	{Compute another radial inverse point}
GTO COS 0 0 4	I020 GTO I004	{Compute another inverse traverse point}
C		{Exit Program Mode = key at lower left}

CHECKSUMS: LBL R : CK=3440

LBL I : CK=6CEB

**INVERSE TRAVERSE PROGRAM
RADIAL INVERSE PROGRAM (Continued)**

For the HP-35s Calculator



TO RUN THE INVERSE TRAVERSE PROGRAM : XEQ I (XEQ COS ENTER)

- ↗ Enter X-Coord. of Beginning Point R/S
- ↑ Enter Y-Coord. of Beginning Point R/S
- ↑ Enter X-Coord. of Ending Point R/S
- ↑ Enter Y-Coord. of Ending Point R/S
- ↑ Read Inverse Distance R/S
- ↑ Read Inverse Azimuth (D.MMSSss) R/S
- ↖ (Next Beginning Point = This Ending Point)

TO RUN THE RADIAL INVERSE PROGRAM : XEQ R (XEQ 7 ENTER)

- Enter X-Coord. of Fixed Point R/S
- Enter Y-Coord. of Fixed Point R/S
- ↗ Enter X-Coord. of Next Point R/S
- ↑ Enter Y-Coord. of Next Point R/S
- ↑ Read Inverse Distance R/S
- ↑ Read Inverse Azimuth (D.MMSSss) R/S
- ↖ (Enter Next Radial Point)

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

AZIMUTH TRAVERSE PROGRAM RADIAL STUB PROGRAM

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
→ XEQ 8	S001 LBL S	
← ▾ 1 2	S002 SF 2	{Flag 2 set = stub mode}
GTO 9 0 0 3	S003 GTO T003	
→ XEQ 9	T001 LBL T	
← ▾ 2 2	T002 CF 2	{Flag 2 clear = traverse mode}
← ▾ 1 4	T003 SF 4	
← ▾ 1 0	T004 SF 0	
XEQ 4 ENTER	T005 XEQ U001	{Get beginning point}
← x◀▶y R/S	T006 INPUT A	{Enter azimuth to new point in D.MMSSss}
← 8	T007 HMS→	
→ RCL EQN	T008 STO Q	
← x◀▶y MODE	T009 INPUT D	{Enter distance to new point}
RCL EQN	T010 RCL Q	
XEQ 4 0 9 7	T011 XEQ U097	{To Rectangular}
RCL + +/-	T012 RCL+ N	
→ RCL 2	T013 STO Y	
x◀▶y	T014 x◁▷y	
RCL + R↓	T015 RCL+ E	
→ RCL 1	T015 STO X	
← R↓ 1	T017 VIEW X	{Read new X-coordinate}
← R↓ 2	T018 VIEW Y	{Read new Y-coordinate}
← ▾ 3 2	T019 FS? 2	{Flag 2 set = stub mode}
GTO 9 0 0 6	T020 GTO T006	{Run another stub traverse point}
GTO 9 0 0 4	T021 GTO T004	{Run another traverse point}
C		{Exit Program Mode = key at lower left}

CHECKSUMS: LBL S : CK=CAAF LBL T : CK=B397

**AZIMUTH TRAVERSE PROGRAM
RADIAL STUB PROGRAM (Continued)**

For the HP-35s Calculator



TO RUN THE AZIMUTH TRAVERSE PROGRAM : XEQ T (XEQ 9 ENTER)

- ↗ Enter X-Coord. of Beginning Point R/S
- ↑ Enter Y-Coord. of Beginning Point R/S
- ↑ Enter Azimuth to New Point (D.MMSSs) R/S
- ↑ Enter Distance to New Point R/S
- ↑ Read X-Coord. of New Point R/S
- ↑ Read Y-Coord. of New Point R/S
- ↖ (Next Beginning Point = This New Point)

TO RUN THE RADIAL STUB PROGRAM : XEQ S (XEQ 8 ENTER)

- Enter X-Coord. of Fixed Point R/S
- Enter Y-Coord. of Fixed Point R/S
- ↗ Enter Azimuth to New Point (D.MMSSs) R/S
- ↑ Enter Distance to New Point R/S
- ↑ Read X-Coord. of New Point R/S
- ↑ Read Y-Coord. of New Point R/S
- ↖ (Enter Next Radial Stub)

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

TRIANGLE PROGRAM – SSS

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ XEQ	C001 LBL C	
➔ x◀▶y 8	C002 INPUT S	
➔ RCL R↓	C003 STO E	
➔ x◀▶y 8	C004 INPUT S	
➔ RCL x◀▶y	C005 STO F	
➔ x◀▶y 8	C006 INPUT S	
➔ RCL i	C007 STO G	
RCL R↓	C008 RCL E	
➔ √x	C009 x ²	
RCL x◀▶y	C010 RCL F	
➔ √x	C011 x ²	
+	C012 +	
RCL i	C013 RCL G	
➔ √x	C014 x ²	
–	C015 –	
XEQ 6 0 1 1	C016 XEQ W011 {Enters # 2}	
RCL X R↓	C017 RCLx E	
RCL X x◀▶y	C018 RCLx F	
÷	C019 ÷	
➔ COS	C020 ACOS	
➔ 8	C021 →HMS	
➔ RCL 6	C022 STO W	
RCL x◀▶y	C023 RCL F	
➔ √x	C024 x ²	
RCL i	C025 RCL G	
➔ √x	C026 x ²	
+	C027 +	
RCL R↓	C028 RCL E	
➔ √x	C029 x ²	
–	C030 –	
XEQ 6 0 1 1	C031 XEQ W011 {Enters # 2}	
RCL X x◀▶y	C032 RCLx F	
RCL X i	C033 RCLx G	
÷	C034 ÷	
➔ COS	C035 ACOS	
➔ 8	C036 →HMS	
➔ RCL 4	C037 STO U	
XEQ 6 0 0 2	C038 XEQ W002 {Enters # 180}	
➔ ENTER	C039 LASTx	

TRIANGLE PROGRAM – SSS (Continued)
For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
–	C040 –	
RCL 6	C041 RCL W	
← 8	C042 HMS→	
–	C043 –	
→ 8	C044 → HMS	
→ RCL 5	C045 STO V	
GTO 4 0 6 6	C046 GTO U066	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL C : CK=3EE3

TO RUN THE SIDE-SIDE-SIDE TRIANGLE PROGRAM : XEQ C (XEQ XEQ ENTER)

Enter Length of First Side	R/S	
Enter Length of Second Side	R/S	
Enter Length of Third Side	R/S	
Read Angle Opposite & First Side	R/S	
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

TRIANGLE PROGRAM – SAS

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ MODE	D001 LBL D	
➡ x◀▶y 8	D002 INPUT S	
➔ RCL R↓	D003 STO E	
➡ x◀▶y R/S	D004 INPUT A	
➔ RCL 6	D005 STO W	
➡ x◀▶y 8	D006 INPUT S	
➔ RCL x◀▶y	D007 STO F	
➔ √x	D008 x ²	
RCL R↓	D009 RCL E	
➔ √x	D010 x ²	
+	D011 +	
XEQ 6 0 1 1	D012 XEQ W011	{Enters # 2}
RCL X R↓	D013 RCLx E	
RCL X x◀▶y	D014 RCLx F	
RCL 6	D015 RCL W	
➡ 8	D016 HMS➔	
COS	D017 COS	
X	D018 x	
-	D019 -	
√x	D020 √x	
➔ RCL i	D021 STO G	
➔ √x	D022 x ²	
RCL x◀▶y	D023 RCL F	
➔ √x	D024 x ²	
+	D025 +	
RCL R↓	D026 RCL E	
➔ √x	D027 x ²	
-	D028 -	
XEQ 6 0 1 1	D029 XEQ W011	{Enters # 2}
RCL X x◀▶y	D030 RCLx F	
RCL X i	D031 RCLx G	
÷	D032 ÷	
➔ COS	D033 ACOS	
➔ 8	D034 ➔HMS	
➔ RCL 4	D035 STO U	
XEQ 6 0 0 2	D036 XEQ W002	{Enters # 180}
➔ ENTER	D037 LASTx	
-	D038 -	
RCL 6	D039 RCL W	

TRIANGLE PROGRAM – SAS (Continued)
For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
← 8	D040 HMS→	
–	D041 –	
→ 8	D042 → HMS	
→ RCL 5	D043 STO V	
GTO 4 0 6 6	D044 GTO U066	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL D : CK=D75D

TO RUN THE SIDE-ANGLE-SIDE TRIANGLE PRGM : XEQ D (XEQ MODE ENTER)

Enter Length of First Side	R/S	
Enter Angle Between (DMS)	R/S	
Enter Length of Second Side	R/S	
Read Angle Opposite & First Side	R/S	
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

TRIANGLE PROGRAM – SAA

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ R↓	E001 LBL E	
➡ x◀▶y 8	E002 INPUT S	
➔ RCL R↓	E003 STO E	
➡ x◀▶y R/S	E004 INPUT A	
➔ RCL 6	E005 STO W	
➡ x◀▶y R/S	E006 INPUT A	
➔ RCL 4	E007 STO U	
➡ 8	E008 HMS➔	
RCL 6	E009 RCL W	
➡ 8	E010 HMS➔	
+	E011 +	
XEQ 6 0 0 2	E012 XEQ W002	{Enters # 180}
x◀▶y	E013 x◁y	
–	E014 –	
➔ 8	E015 ➔HMS	
➔ RCL 5	E016 STO V	
RCL R↓	E017 RCL E	
RCL 4	E018 RCL U	
➡ 8	E019 HMS➔	
SIN	E020 SIN	
÷	E021 ÷	
➔ RCL EQN	E022 STO Q	
RCL 5	E023 RCL V	
➡ 8	E024 HMS➔	
SIN	E025 SIN	
X	E026 x	
➔ RCL x◀▶y	E027 STO F	
RCL EQN	E028 RCL Q	
RCL 6	E029 RCL W	
➡ 8	E030 HMS➔	
SIN	E031 SIN	
X	E032 x	
➔ RCL i	E033 STO G	
GTO 4 0 6 6	E034 GTO U066	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL E : CK=57A9

TRIANGLE PROGRAM – SAA (Continued)
For the HP-35s Calculator



TO RUN THE SIDE-ANGLE-ANGLE TRIANGLE PRGM : XEQ E (XEQ R↓ ENTER)

Enter Length of First Side	R/S	
Enter Angle After (DMS)	R/S	
Enter Next Angle (DMS)	R/S	
Read Angle Opposite & First Side	R/S	
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

TRIANGLE PROGRAM – ASA

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
→ XEQ x◀▶y	F001 LBL F	
← x◀▶y R/S	F002 INPUT A	
→ RCL 5	F003 STO V	
← x◀▶y 8	F004 INPUT S	
→ RCL R↓	F005 STO E	
← x◀▶y R/S	F006 INPUT A	
→ RCL 6	F007 STO W	
← 8	F008 HMS→	
RCL 5	F009 RCL V	
← 8	F010 HMS→	
+	F011 +	
XEQ 6 0 0 2	F012 XEQ W002	{Enters # 180}
x◀▶y	F013 x◁y	
–	F014 –	
→ 8	F015 →HMS	
→ RCL 4	F016 STO U	
RCL R↓	F017 RCL E	
RCL 4	F018 RCL U	
← 8	F019 HMS→	
SIN	F020 SIN	
÷	F021 ÷	
→ RCL EQN	F022 STO Q	
RCL 5	F023 RCL V	
← 8	F024 HMS→	
SIN	F025 SIN	
X	F026 x	
→ RCL x◀▶y	F027 STO F	
RCL EQN	F028 RCL Q	
RCL 6	F029 RCL W	
← 8	F030 HMS→	
SIN	F031 SIN	
X	F032 x	
→ RCL i	F033 STO G	
GTO 4 0 6 6	F034 GTO U066	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL F : CK=93A9

TRIANGLE PROGRAM – ASA (Continued)
For the HP-35s Calculator



TO RUN THE ANGLE-SIDE-ANGLE TRIANGLE PRGM : XEQ F (XEQ x◀▶y ENTER)

Enter Angle Before (DMS)	R/S	
Enter Length of First Side	R/S	
Enter Angle After (DMS)	R/S	
Read Angle Opposite & First Side	R/S	
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

TRIANGLE PROGRAM – SSA

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
→ XEQ i	G001 LBL G	
← ▴ 1 ● 0	G002 SF 10	{● = decimal point}
← x◀▶y 8	G003 INPUT S	
→ RCL R↓	G004 STO E	
← x◀▶y 8	G005 INPUT S	
→ RCL x◀▶y	G006 STO F	
RCL R↓	G007 RCL E	
← MODE 4	G008 x>y?	
→ ⇐ 1	G009 CLx	{If side 1 > side 2, then single solution}
→ RCL 3	G010 STO Z	
← x◀▶y R/S	G011 INPUT A	
→ RCL 4	G012 STO U	
← 8	G013 HMS→	
SIN	G014 SIN	
RCL R↓	G015 RCL E	
÷	G016 ÷	
→ RCL EQN	G017 STO Q	
RCL X x◀▶y	G018 RCLx F	
XEQ 6 0 1 4	G019 XEQ W014	{Enters # 1}
x◀▶y	G020 x<>y	
–	G021 –	
→ MODE 2	G022 x≤0?	
→ RCL 3	G023 STO Z	
→ MODE 3	G024 x<0?	
XEQ 6 0 3 8	G025 XEQ W038	{Message NO SOLUTION}
RCL 3	G026 RCL Z	
→ MODE 3	G027 x<0?	
← XEQ	G028 RTN	
RCL EQN	G029 RCL Q	
RCL X x◀▶y	G030 RCLx F	
→ SIN	G031 ASIN	
→ 8	G032 →HMS	
→ RCL 5	G033 STO V	
← 8	G034 HMS→	
RCL 4	G035 RCL U	
← 8	G036 HMS→	
+	G037 +	
XEQ 6 0 0 2	G038 XEQ W002	{Enters # 180}
x◀▶y	G039 x◀▶y	

TRIANGLE PROGRAM – SSA (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
–	G040 –	
➔ 8	G041 ➔HMS	
➔ RCL 6	G042 STO W	
➔ ENTER	G043 LASTx	
SIN	G044 SIN	
RCL EQN	G045 RCL Q	
÷	G046 ÷	
➔ RCL i	G047 STO G	
RCL 3	G048 RCL Z	
➔ MODE 6	G049 x=0?	{If only one solution exists}
GTO 4 0 6 6	G050 GTO U066	{Display solution and exit}
XEQ 6 0 4 0	G051 XEQ W040	{Message SOLUTION 1}
XEQ 4 0 6 6	G052 XEQ U066	{Display first solution}
XEQ 6 0 0 2	G053 XEQ W002	{Enters # 180}
RCL 5	G054 RCL V	
⬅ 8	G055 HMS➔	
–	G056 –	
➔ 8	G057 ➔HMS	
➔ RCL 5	G058 STO V	
⬅ 8	G059 HMS➔	
RCL 4	G060 RCL U	
⬅ 8	G061 HMS➔	
+	G062 +	
XEQ 6 0 0 2	G063 XEQ W002	{Enters # 180}
x◀▶y	G064 x◁y	
–	G065 –	
➔ 8	G066 ➔HMS	
➔ RCL 6	G067 STO W	
➔ ENTER	G068 LASTx	
SIN	G069 SIN	
RCL EQN	G070 RCL Q	
÷	G071 ÷	
➔ RCL i	G072 STO G	
XEQ 6 0 4 2	G073 XEQ W042	{Message SOLUTION 2}
GTO 4 0 6 6	G074 GTO U066	{Display second solution and exit}
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL G : CK=6A94

TRIANGLE PROGRAM – SSA (Continued)
For the HP-35s Calculator



TO RUN THE SIDE-SIDE-ANGLE TRIANGLE PROGRAM : XEQ G (XEQ i ENTER)

Enter Length of First Side	R/S	
Enter Length of Second Side	R/S	
Enter Angle Opposite First Side (DMS)	R/S	
Prompt "SOLUTION 1"	R/S	{Skip Prompt if Single Solution}
Read Angle Opposite & First Side	R/S	{First or Single Solution}
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program if Single Solution}
Prompt "SOLUTION 2"	R/S	
Read Angle Opposite & First Side	R/S	{Second Solution}
Read Angle Opposite & Second Side	R/S	
Read Angle Opposite & Third Side	R/S	
Read Triangle Area	R/S	{End of Program}

NOTES :

- The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.
- The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

INTERSECTION PROGRAM – LL

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ y^x	L001 LBL L	
⬅ ▲ 2 4	L002 CF 4	
⬅ ▲ 1 0	L003 SF 0	
XEQ 4 ENTER	L004 XEQ U001	{Get line 1 X,Y,AZ}
RCL GTO	L005 RCL B	
⬅ 8	L006 HMS➔	
TAN	L007 TAN	
1/x	L008 1/x	
➔ RCL TAN	L009 STO J	{Slope of line 1 = m1}
⬅ ▲ 2 0	L010 CF 0	
XEQ 4 ENTER	L011 XEQ U001	{Get line 2 X,Y,AZ}
RCL R/S	L012 RCL A	
⬅ 8	L013 HMS➔	
TAN	L014 TAN	
1/x	L015 1/x	
➔ RCL \sqrt{x}	L016 STO K	{Slope of line 2 = m2}
RCL COS	L017 RCL I	
RCL \sqrt{x}	L018 RCL K	
RCL X SIN	L019 RCLx H	
–	L020 –	
➔ RCL EQN	L021 STO Q	{Y-Intercept of line 2 = b2}
RCL TAN	L022 RCL J	
RCL X R↓	L023 RCLx E	
+	L024 +	
RCL – +/-	L025 RCL– N	{Computes b2 – b1}
RCL TAN	L026 RCL J	
RCL – \sqrt{x}	L027 RCL– K	{Computes m1 – m2}
÷	L028 ÷	
➔ RCL 1	L029 STO X	{X = (b2 – b1)/(m1 – m2)}
RCL X \sqrt{x}	L030 RCLx K	
RCL + EQN	L031 RCL+ Q	
➔ RCL 2	L032 STO Y	{Y = m2*X + b2}
⬅ R↓ 1	L033 VIEW X	{Computed intersection X-Coord}
⬅ R↓ 2	L034 VIEW Y	{Computed intersection Y-Coord}
RCL 1	L035 RCL X	
RCL – R↓	L036 RCL– E	
RCL 2	L037 RCL Y	
RCL – +/-	L038 RCL– N	
XEQ 4 0 8 5	L039 XEQ U085	{To Polar}

INTERSECTION PROGRAM – LL (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
x◀▶y	L040 x◁y	
➡ RCL MODE	L041 STO D	
⬅ R↓ MODE	L042 VIEW D	{Distance from first point to intersection}
RCL 1	L043 RCL X	
RCL – SIN	L044 RCL– H	
RCL 2	L045 RCL Y	
RCL – COS	L046 RCL– I	
XEQ 4 0 8 5	L047 XEQ U085	{To Polar}
x◀▶y	L048 x◁y	
➡ RCL MODE	L049 STO D	
⬅ R↓ MODE	L050 VIEW D	{Distance from second point to intersection}
⬅ XEQ	L051 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL L : CK=0183

TO RUN THE LINE–LINE INTERSECTION PROGRAM : XEQ L (XEQ y^x ENTER)

Enter X-Coord. of Point on Line 1	R/S	
Enter Y-Coord. of Point on Line 1	R/S	
Enter Azimuth of Line 1 (D.MMSSss)	R/S	{-999 to compute using a second POT}
Enter X-Coord. of Point on Line 2	R/S	
Enter Y-Coord. of Point on Line 2	R/S	
Enter Azimuth of Line 2 (D.MMSSss)	R/S	{-999 to compute using a second POT}
Read X-Coord. of Intersection	R/S	
Read Y-Coord. of Intersection	R/S	
Read Distance Point 1 to Intersection	R/S	
Read Distance Point 2 to Intersection	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

This calculation is also known as a Bearing-Bearing Intersection.

Register X contains the X-Coord. of the Intersection Point

Register Y contains the Y-Coord. of the Intersection Point

Register D contains the Distance from Point 2 to the Intersection Point

INTERSECTION PROGRAM – LC

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ 1/x	M001 LBL M	
⬅ ▲ 2 4	M002 CF 4	
⬅ ▲ 1 0	M003 SF 0	
XEQ 4 ENTER	M004 XEQ U001	{Get line X,Y,AZ}
RCL GTO	M005 RCL B	
⬅ 8	M006 HMS➔	
TAN	M007 TAN	
1/x	M008 1/x	
➔ RCL √x	M009 STO K	{Slope of line}
➔ √x	M010 x ²	
XEQ 6 0 1 4	M011 XEQ W014	{Enters # 1}
+	M012 +	
➔ RCL R/S	M013 STO A	{Quadratic coefficient A}
RCL COS	M014 RCL I	
RCL √x	M015 RCL K	
RCL X SIN	M016 RCL x H	
–	M017 –	
➔ RCL EQN	M018 STO Q	{Y-Intercept}
⬅ ▲ 2 0	M019 CF 0	
⬅ ▲ 1 4	M020 SF 4	
XEQ 4 ENTER	M021 XEQ U001	{Get radius point X,Y}
⬅ x◀▶y 7	M022 INPUT R	{Get radius}
RCL EQN	M023 RCL Q	
RCL – COS	M024 RCL– I	
RCL X √x	M025 RCLx K	
RCL – SIN	M026 RCL– H	
XEQ 6 0 1 1	M027 XEQ W011	{Enters # 2}
X	M028 x	
➔ RCL GTO	M029 STO B	{Quadratic coefficient B}
RCL SIN	M030 RCL H	
➔ √x	M031 x ²	
RCL COS	M032 RCL I	
➔ √x	M033 x ²	
+	M034 +	
RCL 7	M035 RCL R	
➔ √x	M036 x ²	
–	M037 –	
RCL EQN	M038 RCL Q	
➔ √x	M039 x ²	

INTERSECTION PROGRAM – LC (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
+	M040 +	
XEQ 6 0 1 1	M041 XEQ W011	{Enters # 2}
RCL X EQN	M042 RCLx Q	
RCL X COS	M043 RCLx I	
–	M044 –	
➔ RCL XEQ	M045 STO C	{Quadratic coefficient C}
RCL GTO	M046 RCL B	
➔ \sqrt{x}	M047 x^2	
XEQ 6 0 2 3	M048 XEQ W023	{Enters # 4}
RCL X R/S	M049 RCLx A	
RCL X XEQ	M050 RCLx C	
–	M051 –	
➔ RCL MODE	M052 STO D	{Quadratic discriminant}
➔ MODE 3	M053 $x < 0?$	{If no solution, display and exit}
XEQ 6 0 3 8	M054 XEQ W038	{Message NO SOLUTION}
RCL MODE	M055 RCL D	
➔ MODE 3	M056 $x < 0?$	
➔ XEQ	M057 RTN	
\sqrt{x}	M058 \sqrt{x}	
RCL – GTO	M059 RCL– B	
XEQ 6 0 1 1	M060 XEQ W011	{Enters # 2}
RCL X R/S	M061 RCLx A	
÷	M062 ÷	
➔ RCL 1	M063 STO X	
➔ RCL 4	M064 STO U	
RCL X \sqrt{x}	M065 RCLx K	
RCL + EQN	M066 RCL+ Q	
➔ RCL 2	M067 STO Y	
➔ RCL 5	M068 STO V	
RCL SIN	M069 RCL H	
RCL – 1	M070 RCL– X	
RCL COS	M071 RCL I	
RCL – 2	M072 RCL– Y	
XEQ 4 0 8 5	M073 XEQ U085	{To Polar}
➔ RCL 3	M074 STO Z	
➔ RCL 6	M075 STO W	
RCL R↓	M076 RCL E	
RCL – 1	M077 RCL– X	
RCL +/-	M078 RCL N	
RCL – 2	M079 RCL– Y	
XEQ 4 0 8 5	M080 XEQ U085	{To Polar}
$x \blacktriangleleft \blacktriangleright y$	M081 $x \blacktriangleleft y$	

INTERSECTION PROGRAM – LC (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ RCL 8	M082 STO S	
XEQ 6 0 4 0	M083 XEQ W040	{Message SOLUTION 1}
➡ R↓ 1	M084 VIEW X	{X-Coordinate 1}
➡ R↓ 2	M085 VIEW Y	{Y-Coordinate 1}
➡ R↓ 3	M086 VIEW Z	{Azimuth from CC to INT 1}
➡ R↓ 8	M087 VIEW S	{Distance from POT to INT 1}
RCL MODE	M088 RCL D	
➔ MODE 6	M089 x=0?	{If single solution, exit}
➡ XEQ	M090 RTN	
√x	M091 √x	
+/-	M092 +/-	
RCL - GTO	M093 RCL- B	
XEQ 6 0 1 1	M094 XEQ W011	{Enters # 2}
RCL X R/S	M095 RCLx A	
÷	M096 ÷	
➔ RCL 1	M097 STO X	
RCL X √x	M098 RCLx K	
RCL + EQN	M099 RCL+ Q	
➔ RCL 2	M100 STO Y	
RCL SIN	M101 RCL H	
RCL - 1	M102 RCL- X	
RCL COS	M103 RCL I	
RCL - 2	M104 RCL- Y	
XEQ 4 0 8 5	M105 XEQ U085	{To Polar}
➔ RCL 3	M106 STO Z	
RCL R↓	M107 RCL E	
RCL - 1	M108 RCL- X	
RCL +/-	M109 RCL N	
RCL - 2	M110 RCL- Y	
XEQ 4 0 8 5	M111 XEQ U085	{To Polar}
x◀▶y	M112 x◁▷y	
➔ RCL 8	M113 STO S	
XEQ 6 0 4 2	M114 XEQ W042	{Message SOLUTION 2}
➡ R↓ 1	M115 VIEW X	{X-Coordinate 2}
➡ R↓ 2	M116 VIEW Y	{Y-Coordinate 2}
➡ R↓ 3	M117 VIEW Z	{Azimuth from CC to INT 2}
➡ R↓ 8	M118 VIEW S	{Distance from POT to INT 2}
➡ XEQ	M119 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL M : CK=D8B2

INTERSECTION PROGRAM – LC (Continued)
For the HP-35s Calculator



TO RUN THE LINE–CIRCLE INTERSECTION PROGRAM : XEQ M (XEQ 1/x ENTER)

Enter X-Coord. of Point on Line	R/S	{POT}
Enter Y-Coord. of Point on Line	R/S	{POT}
Enter Azimuth of Line (D.MMSSss)	R/S	{-999 to compute using a second POT}
Enter X-Coord. of Radius Point	R/S	
Enter Y-Coord. of Radius Point	R/S	
Enter Radius of Circle	R/S	
Prompt "SOLUTION 1"	R/S	{First or Single Solution}
Read X-Coord. of Intersection 1	R/S	
Read Y-Coord. of Intersection 1	R/S	
Read Azimuth, Rad. Pt. to Intersection 1	R/S	
Read Distance, POT to Intersection 1	R/S	{End of Program if Single Solution}
Prompt "SOLUTION 2"	R/S	{Second Solution}
Read X-Coord. of Intersection 2	R/S	
Read Y-Coord. of Intersection 2	R/S	
Read Azimuth, Rad. Pt. to Intersection 2	R/S	
Read Distance, POT to Intersection 2	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.
The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

This calculation is also known as a Bearing-Distance Intersection.
Register U contains the X-Coord. of Intersection Point 1
Register V contains the Y-Coord. of Intersection Point 1
Register W contains the Azimuth from the Radius Point to Intersection Point 1
Register X contains the X-Coord. of Intersection Point 2
Register Y contains the Y-Coord. of Intersection Point 2
Register Z contains the Azimuth from the Radius Point to Intersection Point 2

INTERSECTION PROGRAM – CC

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ +/-	N001 LBL N	
⬅ ▲ 2 3	N002 CF 3	
⬅ ▲ 1 4	N003 SF 4	
⬅ ▲ 1 0	N004 SF 0	
XEQ 4 ENTER	N005 XEQ U001	{Get radius point 1 x,y}
⬅ x◀▶y 7	N006 INPUT R	{Get radius length 1}
➔ RCL 8	N007 STO S	
⬅ ▲ 2 0	N008 CF 0	
XEQ 4 ENTER	N009 XEQ U001	{Get radius point 2 x,y}
⬅ x◀▶y 7	N010 INPUT R	{Get radius length 2}
RCL R↓	N011 RCL E	
RCL – SIN	N012 RCL– H	
RCL +/-	N013 RCL N	
RCL – COS	N014 RCL– I	
XEQ 4 0 8 5	N015 XEQ U085	{To Polar}
⬅ 8	N016 HMS➔	
➔ RCL R/S	N017 STO A	{Azimuth CC1 to CC2}
x◀▶y	N018 x◁y	
➔ RCL 9	N019 STO T	{Distance CC1 to CC2}
RCL 8	N020 RCL S	
RCL + 7	N021 RCL+ R	
RCL 9	N022 RCL T	
⬅ MODE 4	N023 x>y?	{Is CC1 to CC2 > R1+R2}
⬅ ▲ 1 3	N024 SF 3	{True = no solution}
⬅ MODE 6	N025 x=y?	{Is CC1 to CC2 = R1+R2}
⬅ ▲ 1 0	N026 SF 0	{True = one solution}
RCL 8	N027 RCL S	
RCL – 7	N028 RCL– R	
➔ +/-	N029 ABS	
RCL 9	N030 RCL T	
⬅ MODE 3	N031 x<y?	{Is CC1 to CC2 < R1-R2 }
⬅ ▲ 1 3	N032 SF 3	{True = no solution}
⬅ ▲ 3 3	N033 FS? 3	{If no solution, display message}
XEQ 6 0 3 8	N034 XEQ W038	{Message NO SOLUTION}
⬅ ▲ 3 3	N035 FS 3?	{If no solution, exit}
⬅ XEQ	N036 RTN	
RCL 8	N037 RCL S	
➔ √x	N038 x ²	
RCL 7	N039 RCL R	
➔ √x	N040 x ²	

INTERSECTION PROGRAM – CC (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
–	N041 –	
RCL 9	N042 RCL T	
→ √x	N043 x ²	
+	N044 +	
XEQ 6 0 1 1	N045 XEQ W011	{Enters # 2}
RCL X 8	N046 RCLx S	
RCL X 9	N047 RCLx T	
÷	N048 ÷	
→ COS	N049 ACOS	
→ RCL MODE	N050 STO D	{Delta1/2}
RCL 8	N051 RCL S	
RCL R/S	N052 RCL A	
RCL + MODE	N053 RCL+ D	
XEQ 4 0 9 7	N054 XEQ U097	{To Rectangular}
RCL + +/-	N055 RCL+ N	
→ RCL 2	N056 STO Y	
→ RCL 5	N057 STO V	
x◀▶y	N058 x<>y	
RCL + R↓	N059 RCL+ E	
→ RCL 1	N060 STO X	
→ RCL 4	N061 STO U	
RCL R↓	N062 RCL E	
RCL – 1	N063 RCL– X	
RCL +/-	N064 RCL N	
RCL – 2	N065 RCL– Y	
XEQ 4 0 8 5	N066 XEQ U085	{To Polar}
→ RCL 3	N067 STO Z	{Azimuth CC1 to Solution 1}
→ RCL TAN	N068 STO J	
XEQ 6 0 4 0	N069 XEQ W040	{Message SOLUTION 1}
← R↓ 1	N070 VIEW X	{X-Coordinate 1}
← R↓ 2	N071 VIEW Y	{Y-Coordinate 1}
← R↓ 3	N072 VIEW Z	{Azimuth CC1 to Solution 1}
RCL SIN	N073 RCL H	
RCL – 1	N074 RCL– X	
RCL COS	N075 RCL I	
RCL – 2	N076 RCL– Y	
XEQ 4 0 8 5	N077 XEQ U085	{To Polar}
→ RCL 3	N078 STO Z	
→ RCL √x	N079 STO K	
← R↓ 3	N080 VIEW Z	{Azimuth CC2 to Solution 1}
← ▴ 3 0	N081 FS 0?	{If single solution, exit}
← XEQ	N082 RTN	

INTERSECTION PROGRAM – CC (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
RCL 8	N083 RCL S	
RCL R/S	N084 RCL A	
RCL – MODE	N085 RCL– D	
XEQ 4 0 9 7	N086 XEQ U097	{To Rectangular}
RCL + +/-	N087 RCL+ N	
➔ RCL 2	N088 STO Y	
x◀▶y	N089 x<>y	
RCL + R↓	N090 RCL+ E	
➔ RCL 1	N091 STO X	
RCL R↓	N092 RCL E	
RCL – 1	N093 RCL– X	
RCL +/-	N094 RCL N	
RCL – 2	N095 RCL– Y	
XEQ 4 0 8 5	N096 XEQ U085	{To Polar}
➔ RCL 3	N097 STO Z	{Azimuth CC1 to Solution 2}
➔ RCL y ^x	N098 STO L	
XEQ 6 0 4 2	N099 XEQ W042	{Message SOLUTION 2}
⬅ R↓ 1	N100 VIEW X	{X-Coordinate 2}
⬅ R↓ 2	N101 VIEW Y	{Y-Coordinate 2}
⬅ R↓ 3	N102 VIEW Z	{Azimuth CC1 to Solution 2}
RCL SIN	N103 RCL H	
RCL – 1	N104 RCL– X	
RCL COS	N105 RCL I	
RCL – 2	N106 RCL– Y	
XEQ 4 0 8 5	N107 XEQ U085	{To Polar}
➔ RCL 3	N108 STO Z	
➔ RCL 1/x	N109 STO M	
⬅ R↓ 3	N110 VIEW Z	{Azimuth CC2 to Solution 2}
⬅ XEQ	N111 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL N : CK=A456

INTERSECTION PROGRAM – CC (Continued)
For the HP-35s Calculator



TO RUN THE CIRCLE–CIRCLE INTERSECTION PRGM : XEQ N (XEQ +/- ENTER)

Enter X-Coord. of Radius Point 1	R/S	
Enter Y-Coord. of Radius Point 2	R/S	
Enter Radius of Circle 1	R/S	
Enter X-Coord. of Radius Point 2	R/S	
Enter Y-Coord. of Radius Point 2	R/S	
Enter Radius of Circle 2	R/S	
Prompt "SOLUTION 1"	R/S	{First or Single Solution}
Read X-Coord. of Intersection 1	R/S	
Read Y-Coord. of Intersection 1	R/S	
Read Azimuth, Rad. Pt. 1 to Intersection 1	R/S	
Read Azimuth, Rad. Pt. 2 to Intersection 1	R/S	{End of Program if Single Solution}
Prompt "SOLUTION 2"	R/S	{Second Solution}
Read X-Coord. of Intersection 2	R/S	
Read Y-Coord. of Intersection 2	R/S	
Read Azimuth, Rad. Pt. 1 to Intersection 2	R/S	
Read Azimuth, Rad. Pt. 2 to Intersection 2	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.
The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

This calculation is also known as a Distance-Distance Intersection.
Register U contains the X-Coord. of Intersection Point 1
Register V contains the Y-Coord. of Intersection Point 1
Register J contains the Azimuth from Radius Point 1 to Intersection Point 1
Register K contains the Azimuth from Radius Point 2 to Intersection Point 1
Register X contains the X-Coord. of Intersection Point 2
Register Y contains the Y-Coord. of Intersection Point 2
Register L contains the Azimuth from Radius Point 1 to Intersection Point 2
Register M contains the Azimuth from Radius Point 2 to Intersection Point 2

HORIZONTAL CURVE PROGRAM

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ SIN	H001 LBL H	
➔ ◀ 1 ● 0	H002 SF 10	{● = decimal point}
➔ ⇐ 2	H003 CLVARS	
XEQ 6 0 0 2	H004 XEQ W002	{Enters # 180}
➔ COS	H005 π	
÷	H006 ÷	
➔ RCL √x	H007 STO K	
XEQ 6 0 4 4	H008 XEQ W044	{Prompts For DELTA}
➔ x◀▶y R/S	H009 INPUT A	{Enter Delta}
➔ 8	H010 HMS➔	
XEQ 6 0 1 1	H011 XEQ W011	{Enters # 2}
÷	H012 ÷	
➔ RCL R/S	H013 STO A	{Delta/2 in A}
➔ x◀▶y MODE	H014 INPUT D	
XEQ 6 0 2 6	H015 XEQ W026	{Enters # 100}
RCL X √x	H016 RCLx K	
RCL MODE	H017 RCL D	{Enter Degree of Curve}
➔ 8	H018 HMS➔	
➔ MODE 4	H019 x>0?	
÷	H020 ÷	
➔ RCL 7	H021 STO R	
➔ MODE 6	H022 x=0?	
➔ x◀▶y 7	H023 INPUT R	{If Degree = 0, Enter Radius}
➔ MODE 6	H024 x=0?	
XEQ SIN 1 4 7	H025 XEQ H147	{If Radius = 0, Compute It}
RCL R/S	H026 RCL A	
➔ MODE 6	H027 x=0?	
XEQ SIN 1 0 5	H028 XEQ H105	{If Delta = 0, Compute It}
RCL R/S	H029 RCL A	
➔ MODE 6	H030 x=0?	
GTO SIN 0 0 3	H031 GTO H003	{If no Valid Delta Angle}
RCL 7	H032 RCL R	
➔ MODE 6	H033 x=0?	
GTO SIN 0 0 3	H034 GTO H003	{If no Valid Radius}
XEQ 6 0 2 6	H035 XEQ W026	{Enters # 100}
RCL X √x	H036 RCLx K	
RCL ÷ 7	H037 RCL÷ R	
➔ 8	H038 ➔HMS	
➔ RCL MODE	H039 STO D	{Degree Of Curve}

HORIZONTAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
RCL R/S	H040 RCL A	
TAN	H041 TAN	
RCL X 7	H042 RCLx R	
➔ RCL 9	H043 STO T	{Tangent Length}
RCL R/S	H044 RCL A	
SIN	H045 SIN	
XEQ 6 0 1 1	H046 XEQ W011	{Enters # 2}
X	H047 x	
RCL X 7	H048 RCLx R	
➔ RCL XEQ	H049 STO C	{Chord Length}
RCL 7	H050 RCL R	
RCL 7	H051 RCL R	
RCL R/S	H052 RCL A	
COS	H053 COS	
X	H054 x	
–	H055 –	
➔ RCL 1/x	H056 STO M	{Mid-Ordinate Distance}
RCL 7	H057 RCL R	
RCL R/S	H058 RCL A	
COS	H059 COS	
÷	H060 ÷	
RCL – 7	H061 RCL– R	
➔ RCL R↓	H062 STO E	{External Distance}
RCL R/S	H063 RCL A	
RCL ÷ √x	H064 RCL÷ K	
RCL X 7	H065 RCLx R	
RCL X 7	H066 RCLx R	
➔ RCL 8	H067 STO S	{Sector Area}
➔ RCL i	H068 STO G	
RCL R/S	H069 RCL A	
COS	H070 COS	
RCL X XEQ	H071 RCLx C	
XEQ 6 0 1 1	H072 XEQ W011	{Enters # 2}
➔ RCL X R/S	H073 STOx A	
÷	H074 ÷	
RCL X 7	H075 RCLx R	
➔ RCL – i	H076 STO– G	{Segment Area}
RCL 7	H077 RCL R	
RCL X 9	H078 RCLx T	
RCL – 8	H079 RCL– S	
➔ RCL x◀▶y	H080 STO F	{Fillet Area}
RCL R/S	H081 RCL A	

HORIZONTAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ 8	H082 →HMS	
← RCL R/S	H083 $x \langle \rangle A$	
RCL $\div \sqrt{x}$	H084 RCL+ K	
RCL X 7	H085 RCLx R	
→ RCL y^x	H086 STO L	{Length Of Curve}
XEQ 6 0 4 4	H087 XEQ W044	{Message DELTA}
← R↓ R/S	H088 VIEW A	
← R↓ MODE	H089 VIEW D	
← R↓ 9	H090 VIEW T	
← R↓ y^x	H091 VIEW L	
← R↓ 7	H092 VIEW R	
← R↓ XEQ	H093 VIEW C	
← R↓ 1/x	H094 VIEW M	
← R↓ R↓	H095 VIEW E	
← R↓ 8	H096 VIEW S	
← R↓ i	H097 VIEW G	
← R↓ $x \blacktriangleleft \blacktriangleright y$	H098 VIEW F	
XEQ 6 0 4 7	H099 XEQ W047	{Prompts PI STA}
RCL - 9	H100 RCL- T	
ENTER	H101 ENTER	
RCL + y^x	H102 RCL+ L	
R/S	H103 STOP	{Display PC and PT Stations}
← XEQ	H104 RTN	
RCL 7	H105 RCL R	{Compute Delta From R}
→ MODE 6	H106 $x=0?$	
← XEQ	H107 RTN	
← $x \blacktriangleleft \blacktriangleright y$ 9	H108 INPUT T	
RCL \div 7	H109 RCL+ R	
→ TAN	H110 ATAN	
→ RCL R/S	H111 STO A	
→ MODE 4	H112 $x>0?$	
← XEQ	H113 RTN	
← $x \blacktriangleleft \blacktriangleright y$ y^x	H114 INPUT L	
RCL X \sqrt{x}	H115 RCLx K	
RCL \div 7	H116 RCL+ R	
XEQ 6 0 1 1	H117 XEQ W011	{Enters # 2}
\div	H118 \div	
→ RCL R/S	H119 STO A	
→ MODE 4	H120 $x>0?$	
← XEQ	H121 RTN	
← $x \blacktriangleleft \blacktriangleright y$ XEQ	H122 INPUT C	
XEQ 6 0 1 1	H123 XEQ W011	{Enters # 2}

HORIZONTAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
÷	H124 ÷	
RCL ÷ 7	H125 RCL÷ R	
➡ SIN	H126 ASIN	
➡ RCL R/S	H127 STO A	
➡ MODE 4	H128 x>0?	
⬅ XEQ	H129 RTN	
⬅ x◀▶y 1/x	H130 INPUT M	
RCL 7	H131 RCL R	
RCL 1/x	H132 RCL M	
-	H133 -	
RCL ÷ 7	H134 RCL÷ R	
➡ COS	H135 ACOS	
➡ RCL R/S	H136 STO A	
➡ MODE 4	H137 x>0?	
⬅ XEQ	H138 RTN	
⬅ x◀▶y R↓	H139 INPUT E	
RCL + 7	H140 RCL+ R	
RCL 7	H141 RCL R	
x◀▶y	H142 x<>y	
÷	H143 ÷	
➡ COS	H144 ACOS	
➡ RCL R/S	H145 STO A	
⬅ XEQ	H146 RTN	
RCL R/S	H147 RCL A	{Compute R from Delta}
➡ MODE 6	H148 x=0?	
⬅ XEQ	H149 RTN	
⬅ x◀▶y 9	H150 INPUT T	
RCL R/S	H151 RCL A	
TAN	H152 TAN	
÷	H153 ÷	
➡ RCL 7	H154 STO R	
➡ MODE 4	H155 x>0?	
⬅ XEQ	H156 RTN	
⬅ x◀▶y y ^x	H157 INPUT L	
RCL X √x	H158 RCLx K	
RCL ÷ R/S	H159 RCL÷ A	
XEQ 6 0 1 1	H160 XEQ W011	{Enters # 2}
÷	H161 ÷	
➡ RCL 7	H162 STO R	
➡ MODE 4	H163 x>0?	
⬅ XEQ	H164 RTN	
⬅ x◀▶y XEQ	H165 INPUT C	

HORIZONTAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
RCL R/S	H166 RCL A	
SIN	H167 SIN	
XEQ 6 0 1 1	H168 XEQ W011	{Enters # 2}
X	H169 x	
÷	H170 ÷	
➔ RCL 7	H171 STO R	
➔ MODE 4	H172 x>0?	
⬅ XEQ	H173 RTN	
⬅ x◀▶y 1/x	H174 INPUT M	
XEQ 6 0 1 4	H175 XEQ W014	{Enters # 1}
RCL R/S	H176 RCL A	
COS	H177 COS	
-	H178 -	
÷	H179 ÷	
➔ RCL 7	H180 STO R	
➔ MODE 4	H181 x>0?	
⬅ XEQ	H182 RTN	
⬅ x◀▶y R↓	H183 INPUT E	
RCL R/S	H184 RCL A	
COS	H185 COS	
1/x	H186 1/x	
XEQ 6 0 1 4	H187 XEQ W014	{Enters # 1}
-	H188 -	
÷	H189 ÷	
➔ RCL 7	H190 STO R	
⬅ XEQ	H191 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL H : CK=822F

HORIZONTAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



TO RUN THE HORIZONTAL CURVE PROGRAM : XEQ H (XEQ SIN ENTER)

Required – Enter at Least One of the Following Three Fields (R/S to Skip) :

Enter the Delta Angle	R/S	{A? D.MMSS}
Enter the Degree of Curve	R/S	{D? D.MMSS – Valid for English Only}
Enter the Curve Radius	R/S	{R? English or Metric}

Optional – Enter One of the Following Fields if Needed :

Enter the Tangent Length	R/S	{T?}
Enter the Curve Length	R/S	{L?}
Enter the Chord Length	R/S	{C?}
Enter the Mid-Ordinate	R/S	{M?}
Enter the External Distance	R/S	{E?}

View the Computed Values :

Read the Delta Angle	R/S	{A= D.MMSS}
Read the Degree of Curve	R/S	{D= D.MMSS – Valid for English Only}
Read the Tangent Length	R/S	{T=}
Read the Curve Length	R/S	{L=}
Read the Curve Radius	R/S	{R=}
Read the Chord Length	R/S	{C=}
Read the Mid-Ordinate	R/S	{M=}
Read the External Distance	R/S	{E=}
Read the Sector Area	R/S	{S=}
Read the Segment Area	R/S	{G=}
Read the Fillet Area	R/S	{F=}
Enter the Station of the PI	R/S	{“PI STA”}
Read the PC and PT Stations	R/S	{End of Program}

NOTES :

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

VERTICAL CURVE (& TANGENT) PROGRAM

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ 5	V001 LBL V	
➔ ◀ 1 ● 0	V002 SF 10	{● = decimal point}
XEQ 6 0 4 9	V003 XEQ W049	{Prompts for PVI STA in feet or meters}
➔ RCL XEQ	V004 STO C	
➔ RCL 9	V005 STO T	
XEQ 6 0 5 1	V006 XEQ W051	{Prompts for PVI ELE or POT if tangent}
➔ RCL MODE	V007 STO D	
➔ RCL 4	V008 STO U	
XEQ 6 0 5 3	V009 XEQ W053	{Prompts for GR IN (%) into PVI}
➔ RCL COS	V010 STO I	
XEQ 6 0 5 5	V011 XEQ W055	{Prompts for GR OUT (%) = G1 if tangent}
➔ RCL E	V012 STO O	
XEQ 6 0 2 6	V013 XEQ W026	{Enters # 100}
➔ RCL ÷ COS	V014 STO+ I	{G1% to decimal}
➔ RCL ÷ E	V015 STO+ O	{G2% to decimal}
XEQ 6 0 5 7	V016 XEQ W057	{Prompt for VC LEN = 0 if tangent grade}
XEQ 6 0 1 1	V017 XEQ W011	{Enters # 2}
÷	V018 ÷	
➔ RCL y ^x	V019 STO L	{=L/2}
➔ RCL - XEQ	V020 STO- C	{=PVC station}
➔ RCL + 9	V021 STO+ T	{=PVT station}
RCL X COS	V022 RCLx I	
➔ RCL - MODE	V023 STO- D	{=PVC elevation}
RCL y ^x	V024 RCL L	
RCL X E	V025 RCLx O	
➔ RCL + 4	V026 STO+ U	{=PVT elevation}
RCL E	V027 RCL O	
RCL - COS	V028 RCL- I	
XEQ 6 0 2 3	V029 XEQ W023	{Enters # 4}
RCL X y ^x	V030 RCLx L	{=2L}
➔ MODE 1	V031 x≠0?	
÷	V032 ÷	
➔ RCL √x	V033 STO K	{(G2-G1)/2L or 0 if tangent}
➔ MODE 6	V034 x=0?	
GTO 5 0 5 3	V035 GTO V053	
RCL COS	V036 RCL I	
XEQ 6 0 1 1	V037 XEQ W011	{Enters # 2}
÷	V038 ÷	
RCL ÷ √x	V039 RCL+ K	

VERTICAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
	V040 +/-	
➡ RCL 1	V041 STO X	
➡ MODE 3	V042 x<0?	
GTO 5 0 5 3	V043 GTO V053	
XEQ 6 0 1 1	V044 XEQ W011	{Enters # 2}
RCL X y ^x	V045 RCLx L	
⬅ MODE 3	V046 x<y?	
GTO 5 0 5 3	V047 GTO V053	
RCL 1	V048 RCL X	
XEQ 5 0 8 0	V049 XEQ V080	
RCL 1	V050 RCL X	
RCL + XEQ	V051 RCL+ C	
R/S	V052 STOP	
XEQ 6 0 5 9	V053 XEQ W059	{Prompts for STA INC = Station Increment}
➡ RCL ()	V054 STO P	
⬅ x◀▶y 8	V055 INPUT S	{Enter station of any point}
XEQ 5 0 6 3	V056 XEQ V063	
RCL 8	V057 RCL S	
RCL R↓	V058 RCL E	
⬅ R↓ R↓	V059 VIEW E	{Read elevation of point}
RCL ()	V060 RCL P	
➡ RCL + 8	V061 STO+ S	
GTO 5 0 5 5	V062 GTO V055	{Enter next station/elevation}
RCL - 9	V063 RCL- T	{Compute elevation from station}
➡ RCL 1	V064 STO X	
RCL X E	V065 RCLx O	
RCL + 4	V066 RCL+ U	
➡ RCL R↓	V067 STO E	
RCL 1	V068 RCL X	
➡ MODE 4	V069 x>0?	
⬅ XEQ	V070 RTN	{On out tangent}
RCL 8	V071 RCL S	
RCL - XEQ	V072 RCL- C	
➡ RCL 1	V073 STO X	
RCL X COS	V074 RCLx I	
RCL + MODE	V075 RCL+ D	
➡ RCL R↓	V076 STO E	
RCL 1	V077 RCL X	
➡ MODE 2	V078 x≤0?	
⬅ XEQ	V079 RTN	{On in tangent}
RCL X COS	V080 RCLx I	
RCL 1	V081 RCL X	

VERTICAL CURVE PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ √x	V082 x ²	
RCL X √x	V083 RCLx K	
+	V084 +	
RCL + MODE	V085 RCL+ D	
→ RCL R↓	V086 STO E	
← XEQ	V087 RTN	{On vertical curve}
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL V : CK=C54B

TO RUN THE VERTICAL CURVE PROGRAM : XEQ V (XEQ 5 ENTER)

Enter the PVI Station	R/S	{Any POT if Computing a Tangent Grade}
Enter PVI Elevation	R/S	{Any POT if Computing a Tangent Grade}
Enter the % Grade into the PVI (G1)	R/S	
Enter the % Grade out of the PVI (G2)	R/S	{= G1 if Computing a Tangent Grade}
Enter the Length of the Vertical Curve	R/S	{Zero if Computing a Tangent Grade}
Read the High or Low Point, If It Exists	R/S	{Elevation in Y- & Station in X-Registers}
Enter a Stationing Increment	R/S	{Prompt is STA INC}
↗ Enter Any Station	R/S	{Prompt is S?}
↑ Read Elevation at the Entered Station	R/S	{Display E=}
↖ Increment for Next Station		

NOTES :

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

AREA BY COORDINATES PROGRAM

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➔ XEQ R/S	A001 LBL A	
⬅ ▲ 1 4	A002 SF 4	
⬅ ▲ 1 0	A003 SF 0	
➔ ⬅ 2	A004 CLVARS	
XEQ 4 ENTER	A005 XEQ U001	{Get First Point X, Y}
⬅ ▲ 2 0	A006 CF 0	
RCL 1	A007 RCL X	{Loop Here Until Next Point = First Point}
➔ RCL 5	A008 STO V	
RCL 2	A009 RCL Y	
➔ RCL 6	A010 STO W	
XEQ 4 ENTER	A011 XEQ U001	{Get Next Point X, Y}
RCL 5	A012 RCL V	
RCL X 2	A013 RCLx Y	
RCL 1	A014 RCL X	
RCL X 6	A015 RCLx W	
-	A016 -	
➔ RCL + x◀▶y	A017 STO+ F	
RCL 1	A018 RCL X	
RCL - 5	A019 RCL- V	
RCL 2	A020 RCL Y	
RCL - 6	A021 RCL- W	
XEQ 4 0 8 5	A022 XEQ U085	{To Polar}
x◀▶y	A023 x◁▷y	
➔ RCL + ()	A024 STO+ P	
RCL 1	A025 RCL X	
RCL R↓	A026 RCL E	
⬅ MODE 1	A027 x≠y?	{If Next X ≠ First X}
GTO R/S 0 0 7	A028 GTO A007	{Loop Back For Next Point}
RCL 2	A029 RCL Y	
RCL +/-	A030 RCL N	
⬅ MODE 1	A031 x≠y?	{If Next Y ≠ First Y}
GTO R/S 0 0 7	A032 GTO A007	{Loop Back For Next Point}
RCL x◀▶y	A033 RCL F	{After Next = First, Show Results}
XEQ 6 0 1 1	A034 XEQ W011	{Enters # 2}
÷	A035 ÷	
➔ RCL x◀▶y	A036 STO F	
XEQ 6 0 2 9	A037 XEQ W029	{Enters # 43,560}
÷	A038 ÷	
➔ RCL R/S	A039 STO A	

AREA BY COORDINATES PROGRAM (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
◀ R↓ x ◀ ▶ y	A040 VIEW F	{Area in Input Units}
◀ R↓ R/S	A041 VIEW A	{Area in Acres if Input Units = Feet}
◀ R↓ ()	A042 VIEW P	{Perimeter in Input Units}
◀ XEQ C	A043 RTN	{Exit Program Mode = key at lower left}

CHECKSUM: LBL A : CK=B275

TO RUN THE AREA BY COORDINATES PROGRAM : XEQ A (XEQ R/S ENTER)

Enter X-Coord. of Beginning Point	R/S	
Enter Y-Coord. of Beginning Point	R/S	
↗ Enter X-Coord. of Next Point	R/S	
↖ Enter Y-Coord. of Next Point	R/S	
↻ Repeats Until Beginning Point Is Re-entered		
Read Area in Square Feet (or Meters)	R/S	{Coordinates are assumed to be in feet.}
Read Area in Acres (Assuming Feet)	R/S	{If units are Meters, ignore this value.}
Read Perimeter	R/S	{End of Program}

NOTES :

The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

HOURS-MINUTES-SECONDS ARITHMETIC HMS PLUS AND HMS MINUS PROGRAMS

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
→ XEQ ()	P001 LBL P	
← 8	P002 HMS→	
x◀▶y	P003 x◁y	
← 8	P004 HMS→	
+	P005 +	
XEQ 6 0 0 5	P006 XEQ W005	{Enters # 360}
← TAN 3	P007 RMDR	{Adjusts so that $0 \leq \text{Azimuth} < 360$ }
→ 8	P008 →HMS	
← XEQ	P009 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL P : CK=68B2

TO RUN THE HMS+ PROGRAM :

Enter the first angle in DDD.MMSSss
 Enter the angle to add in DDD.MMSSss
 Read the sum of the angles in DDD.MMSSss

ENTER
XEQ P [XEQ () ENTER]

TO RUN THE HMS- PROGRAM :

Enter the first angle in DDD.MMSSss
 Enter the angle to subtract in DDD.MMSSss
 Read the difference of the angles in DDD.MMSSss

ENTER
+/- XEQ P [XEQ () ENTER]

NOTES :

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

POLAR → RECTANGULAR (y,x → θ ,r) FUNCTION
RECTANGULAR → POLAR (θ ,r → y,x) FUNCTION
 For the HP-35s Calculator



P→R PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
➔ XEQ TAN	J001 LBL J	
➔ 8	J002 HMS➔	
XEQ 4 0 9 7	J003 XEQ U097	{To Rectangular}
➔ XEQ C	J004 RTN	{Exit Program Mode = key at lower left}

CHECKSUM: LBL J : CK=BB1B

TO RUN THE POLAR→RECTANGULAR FUNCTION :

Enter the Distance	ENTER
Enter the Azimuth (D.MMSSss)	XEQ J [XEQ TAN ENTER]
Read the X-Coordinate difference	{X-Difference in the Y-Register}
Read the Y-Coordinate difference	{Y-Difference in the X-Register}

NOTE : The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.

R→P PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➔ R/S		{Enter Program Mode}
GTO ••	PRGM TOP	{•• = decimal point twice}
➔ XEQ \sqrt{x}	K001 LBL K	
XEQ 4 0 8 5	K002 XEQ U085	{To Polar}
XEQ 6 0 0 2	K003 XEQ W002	{Enters # 180}
GTO () ENTER C	K004 GTO P001	{Exit Program Mode = key at lower left}

CHECKSUM: LBL K : CK=B432

TO RUN THE RECTANGULAR→POLAR FUNCTION :

Enter the X-Coordinate difference	ENTER
Enter the Y-Coordinate difference	XEQ K [XEQ \sqrt{x} ENTER]
Read the resulting distance	{Distance in the Y-Register}
Read the resulting azimuth in DDD.MMSSss	{Azimuth in the X-Register}

NOTES :

- The **UTILITY SUBROUTINES** (LBL U) must be stored in the HP-35s.
- The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.
- The **HMS+ Program** (LBL P) must be stored in the HP-35s.

RATIO PROGRAM (Linear Proportion of X&Y Values)

For the HP-35s Calculator



PROGRAM ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
→ XEQ E	O001 LBL O	
← ▴ 1 4	O002 SF 4	
→ ⇐ 4	O003 CLΣ	
← ▴ 1 0	O004 SF 0	
XEQ 4 ENTER	O005 XEQ U001	{Get beginning X-Value & Y-Value}
Σ+	O006 Σ+	
← ▴ 2 0	O007 CF 0	
XEQ 4 ENTER	O008 XEQ U001	{Get ending X-Value & Y-Value}
Σ+	O009 Σ+	
← - 4	O010 m	
XEQ 6 0 1 4	O011 XEQ W014	{Enters # 1}
R/S	O012 STOP	{Display X:Y Ratio}
RCL R↓	O013 RCL E	
→ RCL 1	O014 STO X	
XEQ 6 0 5 9	O015 XEQ W059	{Prompts for STA INC = Station Increment}
→ RCL ()	O016 STO P	
RCL ()	O017 RCL P	
→ RCL + 1	O018 STO+ X	
← x◀▶y 1	O019 INPUT X	{Enter X-Value to Compute}
RCL 1	O020 RCL X	
← - 1	O021 \hat{x}	
R/S	O022 STOP	{Display Entered X : Computed Y}
GTO E 0 1 7	O023 GTO O017	{Get Next X-Value}
← x◀▶y 2	O024 INPUT Y	{Enter Y-Value to Compute}
← - 2	O025 \hat{y}	
RCL 2	O026 RCL Y	
R/S	O027 STOP	{Display Computed X : Entered Y}
GTO E 0 2 4	O028 GTO O024	{Get Next Y-Value}
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL O : CK=4E69

RATIO PROGRAM (Continued)

For the HP-35s Calculator



TO RUN THE RATIO PROGRAM : XEQ O (XEQ E ENTER)

Enter X-Value of Beginning Point	R/S	{Typically the first station}
Enter Y-Value at Beginning Point	R/S	{Value at start of taper, super transition, etc.}
Enter X-Value of Ending Point	R/S	{Typically the last station}
Enter Y-Value at Ending Point	R/S	{Value at end of taper, super transition, etc.}
Displays Ratio X:Y	R/S	{Goes directly into Y-Value computation}

Y-Value computation

Enter an increment for the X-Value	R/S	{Facilitates computation at regular intervals}
➤ X-Value at which to compute Y-Value	R/S	{Accept incremented value or enter another}
⬅ Read X-Value and computed Y-Value	R/S	{X-Value above and Y-Value below}

X-Value computation

➤ Y-Value for which to compute X-Value	R/S	
⬅ Read computed X-Value and Y-Value	R/S	{X-Value above and Y-Value below}

NOTES :

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

The user can switch to **X-Value computation** by keying **XEQ E 0 2 4**.

The user can switch to **Y-Value computation** by keying **XEQ E 0 1 5**.

UTILITY SUBROUTINES

For the HP-35s Calculator



LABEL U ENTRY SEQUENCE (Coordinate Entry, Triangle Result, Etc.)

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
→ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
→ XEQ 4	U001 LBL U	
← ▲ 1 ● 0	U002 SF 10	{● = decimal point}
← ▲ 1 1	U003 SF 1	
← ▲ 3 0	U004 FS? 0	
← ▲ 2 1	U005 CF 1	
← ▲ 3 0	U006 FS? 0	
XEQ 6 0 3 2	U007 XEQ W032	{Prompts for FIRST}
← ▲ 3 1	U008 FS? 1	
XEQ 6 0 3 5	U009 XEQ W035	{Prompts for NEXT}
← x◀▶y 1	U010 INPUT X	{Enter first/next X-coordinate}
→ RCL SIN	U011 STO H	
← ▲ 3 0	U012 FS? 0	
→ RCL R↓	U013 STO E	{If first point}
← x◀▶y 2	U014 INPUT Y	{Enter first/next Y-coordinate}
→ RCL COS	U015 STO I	
← ▲ 3 0	U016 FS? 0	
→ RCL +/-	U017 STO N	{If first point}
← ▲ 3 4	U018 FS? 4	
← XEQ	U019 RTN	{Return if X/Y only}
XEQ 6 0 1 7	U020 XEQ W017	{Enters # -999}
→ RCL R/S	U021 STO A	
← x◀▶y R/S	U022 INPUT A	{Enter first/next azimuth}
← MODE 1	U023 x≠y?	{Leave -999 to calculate azimuth}
GTO 4 0 4 2	U024 GTO U042	{If azimuth was entered}
← x◀▶y 1	U025 INPUT X	{Enter X-Coordinate of point on line}
← x◀▶y 2	U026 INPUT Y	{Enter Y-Coordinate of point on line}
→ ⇐ 1	U027 CLx	
→ RCL R/S	U028 STO A	{Set Azimuth to zero for limits test}
RCL SIN	U029 RCL H	
RCL - 1	U030 RCL- X	
→ MODE 6	U031 x=0?	{If X1=X2 then}
GTO 4 0 4 2	U032 GTO U042	{Leave azimuth = 0}
XEQ 6 0 0 8	U033 XEQ W008	{Enters # 90}
→ RCL R/S	U034 STO A	{Set Azimuth to 90 for limits test}
→ ⇐ 1	U035 CLx	
RCL COS	U036 RCL I	
RCL - 2	U037 RCL- Y	
→ MODE 6	U038 x=0?	{If Y1=Y2 then}
GTO 4 0 4 2	U039 GTO U042	{Leave azimuth = 90}

UTILITY SUBROUTINES (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
XEQ 4 0 8 5	U040 XEQ U085	{To Polar}
➔ RCL R/S	U041 STO A	
RCL R/S	U042 RCL A	{Handle entered azimuth}
⬅ 8	U043 HMS➔	
XEQ 6 0 0 2	U044 XEQ W002	{Enters # 180}
⬅ MODE 4	U045 x>y?	
➔ ⇐ 1	U046 CLx	{If 180 is larger than azimuth, subtract 0}
-	U047 -	{Otherwise subtract 180}
➔ RCL R/S	U048 STO A	
XEQ 6 0 2 0	U049 XEQ W020	{Enters # 1E -9}
➔ RCL E	U050 STO O	
XEQ 6 0 0 8	U051 XEQ W008	{Enters # 90}
RCL R/S	U052 RCL A	
⬅ MODE 6	U053 x=y?	{If azimuth = 90, subtract a small number}
RCL - E	U054 RCL- O	
➔ RCL R/S	U055 STO A	
RCL R/S	U056 RCL A	
➔ MODE 6	U057 x=0?	{If azimuth = 0, substitute a small number}
RCL E	U058 RCL O	
➔ RCL R/S	U059 STO A	
RCL R/S	U060 RCL A	
➔ 8	U061 ➔HMS	
➔ RCL R/S	U062 STO A	
⬅ ▴ 3 0	U063 FS? 0	
➔ RCL GTO	U064 STO B	{If first azimuth}
⬅ XEQ	U065 RTN	
RCL 4	U066 RCL U	{TRIANGLES - angle opposite first side}
RCL R↓	U067 RCL E	{Length of first side}
R/S	U068 STOP	
RCL 5	U069 RCL V	{Angle opposite second side}
RCL x◀▶y	U070 RCL F	{Length of second side}
R/S	U071 STOP	
RCL 6	U072 RCL W	{Angle opposite third side}
RCL i	U072 RCL G	{Length of third side}
R/S	U074 STOP	
RCL 6	U075 RCL W	
⬅ 8	U076 HMS➔	
SIN	U077 SIN	
XEQ 6 0 1 1	U078 XEQ W011	{Enters # 2}
÷	U079 ÷	
RCL X R↓	U080 RCLx E	
RCL X x◀▶y	U081 RCLx F	

UTILITY SUBROUTINES (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➡ RCL R/S	U082 STO A	
⬅ R↓ R/S	U083 VIEW A	{Area of the triangle}
⬅ XEQ	U084 RTN	
x◀▶y	U085 x◁y	{To Polar function}
i	U086 i	
X	U087 x	
+	U088 +	
⬅ i	U089 ARG	
➡ ENTER	U090 LASTx	
➡ +/-	U091 ABS	
x◀▶y	U092 x◁y	
XEQ 6 0 0 2	U093 XEQ W002	{Enters # 180}
+	U094 +	
➡ 8	U095 →HMS	
⬅ XEQ	U096 RTN	
COS	U097 COS	{To Rectangular function}
➡ ENTER	U098 LASTx	
SIN	U099 SIN	
XEQ 6 0 6 1	U100 XEQ W061	
X	U101 x	
x◀▶y	U102 x◁y	
➡ ENTER	U103 LASTx	
X	U104 x	
⬅ XEQ	U105 RTN	
C		{Exit Program Mode = key at lower left}

CHECKSUM: LBL U : CK= 0366

NOTES :

The **NUMERIC AND PROMPTING DATA** (LBL W) must be stored in the HP-35s.

NUMERIC AND PROMPTING DATA

For the HP-35s Calculator



LABEL W ENTRY SEQUENCE

<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
➡ ← 2		{Not in Program Mode}
➡ R/S		{Enter Program Mode}
GTO ●●	PRGM TOP	{●● = decimal point twice}
➡ XEQ 6	W001 LBL W	
ENTER	W002 ENTER	
1 8 0	W003 180	
➡ XEQ	W004 RTN	
ENTER	W005 ENTER	
3 6 0	W006 360	
➡ XEQ	W007 RTN	
ENTER	W008 ENTER	
9 0	W009 90	
➡ XEQ	W010 RTN	
ENTER	W011 ENTER	
2	W012 2	
➡ XEQ	W013 RTN	
ENTER	W014 ENTER	
1	W015 1	
➡ XEQ	W016 RTN	
ENTER	W017 ENTER	
9 9 9 +/-	W018 -999	
➡ XEQ	W019 RTN	
ENTER	W020 ENTER	
1 E 9 +/-	W021 1E -9	{A small number to avoid trig errors}
➡ XEQ	W022 RTN	
ENTER	W023 ENTER	
4	W024 4	
➡ XEQ	W025 RTN	
ENTER	W026 ENTER	
1 0 0	W027 100	
➡ XEQ	W028 RTN	
ENTER	W029 ENTER	
4 3 5 6 0	W030 43560	{Square feet per acre}
➡ XEQ	W031 RTN	
EQN RCL x◀▶y	RCL COS RCL 7 RCL 8 RCL 9 ENTER	
	W032 FIRST	
➡ x◀▶y	W033 PSE	
➡ XEQ	W034 RTN	
EQN RCL +/- RCL R↓ RCL 1 RCL 9 ENTER		
	W035 NEXT	
➡ x◀▶y	W036 PSE	
➡ XEQ	W037 RTN	

NUMERIC AND PROMPTING DATA (Continued)

For the HP-35s Calculator



<u>Keystrokes</u>	<u>Display</u>	<u>Comments</u>
EQN RCL +/- RCL E → 0 RCL 8 RCL E RCL y ^x RCL 4 RCL 9 RCL COS RCL E RCL +/- ENTER	W038 NO SOLUTION	
← XEQ	W039 RTN	
EQN RCL 8 RCL E RCL y ^x RCL 4 RCL 9 RCL COS RCL E RCL +/- → 0 1 ENTER	W040 SOLUTION 1	
← XEQ	W041 RTN	
EQN RCL 8 RCL E RCL y ^x RCL 4 RCL 9 RCL COS RCL E RCL +/- → 0 2 ENTER	W042 SOLUTION 2	
← XEQ	W043 RTN	
EQN RCL MODE RCL R↓ RCL y ^x RCL 9 RCL R/S ENTER	W044 DELTA	
→ x◀▶y	W045 PSE	
← XEQ	W046 RTN	
EQN RCL () RCL COS → 0 RCL 8 RCL 9 RCL R/S ENTER	W047 PI STA	
← XEQ	W048 RTN	
EQN RCL () RCL 5 RCL COS → 0 RCL 8 RCL 9 RCL R/S ENTER	W049 PVI STA {PVI station in feet or meters}	
← XEQ	W050 RTN	
EQN RCL () RCL 5 RCL COS → 0 RCL R↓ RCL y ^x RCL R↓ ENTER	W051 PVI ELE {PVI elevation or POT if tangent}	
← XEQ	W052 RTN	
EQN RCL i RCL 7 → 0 RCL COS RCL +/- ENTER	W053 GR IN {G1% into PVI}	
← XEQ	W054 RTN	
EQN RCL i RCL 7 → 0 RCL E RCL 4 RCL 9 ENTER	W055 GR OUT {G2% out of PVI = G1 if tangent}	
← XEQ	W056 RTN	
EQN RCL 5 RCL XEQ → 0 RCL y ^x RCL R↓ RCL +/- ENTER	W057 VC LEN {Zero if tangent grade}	
← XEQ	W058 RTN	
EQN RCL 8 RCL 9 RCL R/S → 0 RCL COS RCL +/- RCL XEQ ENTER	W059 STA INC {Station increment}	
← XEQ	W060 RTN	
← ▲ 2 • 0	W061 CF 10	{• = decimal point}
EQN R↓ 3 ENTER	W062 REGZ	
← ▲ 1 • 0	W063 SF 10	{• = decimal point}
← XEQ	W064 RTN	

CHECKSUM: LBL W : CK=875E



HP-35s CALCULATOR PROGRAMS

Mn/DOT Office of Land Management

Surveys Research & Support Unit



NOTES ON THE 11-20-2008 VERSION:

This is the first version for the HP-35s. It was adapted from the latest version for the HP-33s.

The HP-35s lacks the **R→P** and **P→R** functions found on almost every other HP calculator. This listing includes substitutes for those functions. The inverse and traverse programs provide other ways to convert between rectangular and polar notations.

The HP-35s has several known bugs, among which is a problem with the checksums when entering numbers or equations. All such entries are collected into LBL W (**NUMERIC AND PROMPTING DATA**) above. Your checksum for LBL W may not match the checksum shown, but **all other checksums should match**. Keying in LBL W before any other labels may help to make the checksum match. Be sure to enter the keystrokes exactly, and the programs should work properly, even if the checksum does not match.

NOTES ON THE 02-03-2009 VERSION:

The 11-20-2008 version contained several typographical errors. Please note the corrections on lines: G010, G023, G026, G048, M066, N011-N014, H025, V013, W038, W053, and the LBL V checksum.

A new Ratio Program, LBL O, has been added.

NOTES ON THE 04-21-2009 VERSION:

The 02-03-2009 version contained an error in LBL J. To correct the problem, insert one new line as shown in the listing. Note the new checksum. Thank you to Gerald Jennings for the catch.

NOTES ON THE 05-28-2009 VERSION:

All previous versions contained a typographical error on line W057. The 02-03-2009 version fixed a similar problem on line W053, but missed this one.

NOTES ON THE 04-01-2010 VERSION:

This version differs from the May 28, 2009 version only in Label U. It adds 9 lines that check for $X1=X2$ or $Y1=Y2$ when entering a second point to compute an azimuth. Some other checksums changed, but the calculator should take care of them when the new lines are added.